

6. Câu lệnh with

Trong chương trình Xep_loai ở §13 ta thấy sau câu lệnh *for-do* mỗi khi truy cập đến một trường của biến bản ghi *Lop[i]*, ta đều phải viết:

```
Lop[i].<tên trường>
```

Như vậy, cần xử lý bao nhiêu trường thì phải viết kèm bấy nhiêu lần *Lop[i]*. Pascal cung cấp lệnh *with* để tham chiếu đến các trường của biến bản ghi một cách ngắn gọn như sau:

```
with <tên biến bản ghi> do  
    <câu lệnh>;
```

Trường (không kèm tên *biến bản ghi*) trong *câu lệnh* được hiểu là trường của *biến bản ghi* đó.

Ví dụ

Ta có thể viết lại chương trình Xep_loai ở §13 như sau:

```
program Xep_loai;
uses crt;
const Max = 45;
type HocSinh = record
    HoTen: string[30];
    NgaySinh: string[10];
    DiaChi: string[50];
    Toan,Van: real;
    XepLoai: char;
end;
var Lop: array[1..Max] of HocSinh;
    N,i: byte;
begin
    clrscr;
    write('Số lượng học sinh trong lớp N = ');
    readln(N);
    for i:=1 to N do
        with lop[i] do
            begin
                writeln('Các số liệu về học sinh thu ',i,' : ');
                write('Họ và tên: '); readln(HoTen);
                write('Ngày sinh: '); readln(NgaySinh);
                write('Địa chỉ: '); readln(DiaChi);
                write('Điểm Toán: '); readln(Toan);
                write('Điểm Văn: '); readln(Van);
                if Toan+Van >=18 then XepLoai:= 'A';
                if (Toan+Van>=14) and (Toan+Van<18) then XepLoai:='B';
                if (Toan+Van>=10) and (Toan+Van<14) then XepLoai:='C';
                if Toan+Van<10 then XepLoai:= 'D';
            end;
        end;
    clrscr;
    writeln('Danh sách xếp loại học sinh trong lớp:');
    for i:= 1 to N do
        writeln(lop[i].HoTen:30, ' -Xếp loại:',lop[i].XepLoai);
    readln
end.
```